

Formalising Proj Construction in Lean

Jujian Zhang   

Department of Mathematics, Imperial College London

Abstract

Many objects of interest in mathematics can be studied both analytically and algebraically, while at the same time, it is known that analytic geometry and algebraic geometry generally does not behave the same. However, the famous GAGA theorem asserts that for projective varieties, analytic and algebraic geometries are closely related; proof of the Fermat last theorem, for example, use this technique to transport between the two worlds [13]. A crucial step of proving GAGA is to calculate cohomology of projective spaces [12, 8], thus I formalise Proj construction for any \mathbb{N} -graded R -algebra A as a starting point to the GAGA theorem and projective n -space is constructed as $\text{Proj } A[X_0, \dots, X_n]$. This would be the first family of non-affine schemes formalised in any theorem prover.

2012 ACM Subject Classification Theory of computation \rightarrow Logic and verification; Mathematics of computing \rightarrow Topology

Keywords and phrases Lean, formalisation, algebraic geometry, scheme, Proj construction, projective geometry

Digital Object Identifier 10.4230/LIPIcs.CVIT.2016.23

Supplementary Material *Software (Source Code)*: <https://github.com/leanprover-community/mathlib/pull/18138/commits/00c4b0918a2c7a8b62291581b0e1eddf2357b5be>

Funding Jujian Zhang: Schrödinger Scholarship Scheme

Acknowledgements I want to thank Eric Wieser for his contribution and suggestion in formalising homogeneous ideal and homogeneous localisation; Andrew Yang and Junyan Xu for their review and suggestion of my often badly written code; Kevin Buzzard for suggesting this project and all the contributors to `mathlib` for otherwise this project would not have been possible.

1 Introduction

Algebraic geometry concerns polynomials and analytic geometry concerns holomorphic functions. Though all polynomials are holomorphic, the converse is not true; thus many analytic objects are not algebraic, for example $\{x \in \mathbb{C} \mid \sin(x) = 0\}$ can not be defined as zero locus of polynomials in one variable, for polynomials always have only finite number of zeros. However, for projective varieties over \mathbb{C} , the categories of algebraic and analytic coherent sheaves are equivalent; a consequence for this statement is that all closed analytic subspace of projective n -space \mathbb{P}_n is also algebraic [13, 4]. A crucial step of proving the above statement is to consider cohomology of projective n -space \mathbb{P}_n [12].

While one can define \mathbb{P}_n over \mathbb{C} without consideration of other projective varieties, it would be more fruitful to formalise Proj construction as a **scheme** and recover \mathbb{P}_n as $\text{Proj } \mathbb{C}[X_0, \dots, X_n]$, since, among other reasons, by considering different base rings, one obtain different projective varieties, for example, for any homogeneous polynomials f_1, \dots, f_k , $\text{Proj} \left(\frac{\mathbb{C}[X_0, \dots, X_n]}{(f_1, \dots, f_k)} \right)$ defines a projective hypersurface over \mathbb{C} .

In this paper I describe a formal construction of $\text{Proj } A$ in the Lean3 theorem prover [7] by closely following [9, Chapter II]. The formal construction uses various results from the Lean mathematical library `mathlib`, most notably the graded algebra and `Spec` construction; this project has been partly accepted into `mathlib` already while the remaining part is still



© Jujian Zhang;
licensed under Creative Commons License CC-BY 4.0

42nd Conference on Very Important Topics (CVIT 2016).

Editors: John Q. Open and Joan R. Access; Article No. 23; pp. 23:1–23:16



Leibniz International Proceedings in Informatics

LIPICS Schloss Dagstuhl – Leibniz-Zentrum für Informatik, Dagstuhl Publishing, Germany

undergoing a review process. The code discussed in this paper can be found on GitHub¹. I have freely used the axiom of choice and the law of excluded middle throughout this project since the rest of `mathlib` freely use classical reasoning as well; consequently, the final construction is not computable.

As previously mentioned, Proj construction heavily depends on graded algebra and Spec construction. A detailed description of graded algebra in Lean and `mathlib` as well as a comparison of graded algebra with that in other theorem provers can be found in [17]; for my purpose, I have chosen to use internal grading for any graded ring $A \cong \bigoplus \mathcal{A}_i$ so that the result of the construction is about homogeneous prime ideals of A directly instead of $\bigoplus_i \mathcal{A}_i$. The earliest complete Spec construction in Lean can be found in [2] where the construction followed a “sheaf-on-a-basis” approach from [14, Section 01HR], however, it differs significantly from the Spec construction currently found in `mathlib` where the construction follows [9, Chapter II]; for this reason, I have also chosen to follow the definition in [9, Chapter II] while hand-waving part (which is almost the whole proof) was made to be explicit. Some other theorem provers also have or partially have Spec construction: in Isabelle/HOL, Spec is formalised by using locales and rewriting topology and ring theory part of the existing library in [1], however, the category of scheme is yet to be formalized; an early formalisation of Spec in Coq can be found in [3] and a definition of scheme in general can be found in its `UniMath` library [16]; due to homotopy type theory of `Agda`, only a partial formalisation of Spec construction can be found in [11]. Though some theorem provers have defined a general scheme, I could not find any concrete construction of a scheme other than Spec of a ring².

After explaining the mathematical details involved in Proj construction in Section 2, Lean code will be provided and explained in Section 3. For typographical reasons, some code of formalisation will be omitted and marked as `sorry` or `_` and some code presented in this paper is pseudocode that closely resembles the actual code but with notations and names altered to make it more readable and presentable.

2 Mathematical details

In this section, certain familiarity with basic ring theory, topology and category theory will be assumed. In Sections 2.1 and 2.2, definition of a scheme is explained in detail; Spec construction will also be briefly explained to fix the mathematical approach used in `mathlib`. Then by following the definition of a scheme step by step, Proj construction will be explained in Section 2.3.

2.1 Sheaves and Locally Ringed Spaces

Let X be a topological space and $\mathcal{O}pens(X)$ be the category of open subsets of X .

► **Definition 1** (Presheaves [10]). *Let C be a category, a C -valued presheaf \mathcal{F} on X is a functor $\mathcal{O}pens(X)^{op} \Rightarrow C$. Morphisms between C -valued presheaves \mathcal{F}, \mathcal{G} are natural transformations. The category thus formed is denoted as $\mathfrak{Psh}(X, C)$.*

In this paper, the category of interest is the category of presheaves of rings $\mathfrak{Psh}(X, \mathfrak{Ring})$. More explicitly, a presheaf of rings \mathcal{F} assigns each open subset $U \subseteq X$ with a ring $\mathcal{F}(U)$ whose elements are called sections on U and for any open subsets $U \subseteq V \subseteq X$, \mathcal{F} assigns a

¹ url: <https://github.com/leanprover-community/mathlib/pull/18138/>

² In this paper, all rings are assumed to be unital and commutative.

ring homomorphism $\mathcal{F}(V) \rightarrow \mathcal{F}(U)$ often denoted as res_U^V or simply with a vertical bar $s|_U$ (a section s on V restricted to U). Examples of presheaf of rings are abundant: considering open subsets of \mathbb{C} , $U \mapsto \{(\text{continuous, holomorphic}) \text{ functions on } U\}$ with the natural restriction map defines presheaf of rings. In these examples, compatible sections on different open subsets can be glued together to form bigger sections on the union of the said open subsets; this property can be generalized to arbitrary categories:

► **Definition 2** (Sheaves [10, 14]). A presheaf $\mathcal{F} \in \mathfrak{Psh}(X, C)$ is said to be a sheaf if for any open covering of open set $U = \bigcup_i U_i \subseteq X$, the following diagram is an equalizer

$$\mathcal{F}(U) \xrightarrow{(\text{res}_{U_i}^U)} \prod_i \mathcal{F}(U_i) \xrightarrow[\left(\text{res}_{U_i \cap U_j}^{U_j}\right)]{\left(\text{res}_{U_i \cap U_j}^{U_i}\right)} \prod_{i,j} \mathcal{F}(U_i \cap U_j).$$

The category of sheaves $\mathfrak{Sh}(X, C)$ is the full subcategory of the category of presheaves satisfying the sheaf condition.

► **Definition 3** (Locally Ringed Space [14, 9]). If \mathcal{O}_X is a sheaf on X , then the pair (X, \mathcal{O}_X) is called a ringed space; a morphism between two ringed space (X, \mathcal{O}_X) and (Y, \mathcal{O}_Y) is a pair (f, ϕ) such that $f : X \rightarrow Y$ is continuous and $\phi : \mathcal{O}_Y \rightarrow f_*\mathcal{O}_X$ is a morphism of sheaves where $f_*\mathcal{O}_X \in \mathfrak{Sh}(Y)$ assigns $V \subseteq Y$ to $\mathcal{O}_X(f^{-1}(V))$. A locally ringed space (X, \mathcal{O}_X) is ringed space such that for any $x \in X$, its stalk $\mathcal{O}_{X,x}$ is a local ring where $\mathcal{O}_{X,x} = \text{colim}_{x \in U \in \text{Opens } X} \mathcal{O}_X(U)$; a morphism between two locally ringed spaces (X, \mathcal{O}_X) and (Y, \mathcal{O}_Y) is a morphism (f, ϕ) of ringed space such that for any $x \in X$ the ring homomorphism induced on stalk $\phi_x : \mathcal{O}_{Y,f(x)} \rightarrow \mathcal{O}_{X,x}$ is local.

Following from the previous definitions, if \mathcal{O}_X is a presheaf and $U \subseteq X$ is an open subset, then there is a presheaf $\mathcal{O}_X|_U$ on U by assigning every open subset V of U to $\mathcal{O}_X(V)$, this is called restricting a presheaf; sheaves, ringed spaces and locally ringed spaces can also be similarly restricted.

2.2 Definition of Affine Scheme and Scheme

Spec construction

Let R be a ring and $\text{Spec } R$ denote the set of prime ideals of R . Then for any subset $s \subseteq R$, its zero locus is defined as $\{\mathfrak{p} \mid s \subseteq \mathfrak{p}\}$. These zero loci can be considered as closed subsets of $\text{Spec } R$, the topology thus formed is called the Zariski topology. Then a sheaf of rings on $\text{Spec } R$ can be defined by assign $U \subseteq \text{Spec } R$ to the ring

$$\left\{ s : \prod_{x \in U} R_x \mid s \text{ is locally a fraction} \right\},$$

where s is locally a fraction if and only if for any prime ideal $x \in U$, there is always an open subset $x \in V \subseteq U$ and $a, b \in R$ such that for any prime ideal $y \in V$, $b \notin y$ and $s(y) = \frac{a}{b}$. This sheaf \mathcal{O} is called the structure sheaf of $\text{Spec } R$. $(\text{Spec } R, \mathcal{O})$ is a locally ringed space because for any prime ideal $x \subseteq R$, $\mathcal{O}_x \cong A_x$ [9].

Scheme

► **Definition 4** (Scheme). A locally ringed space (X, \mathcal{O}_X) is said to be a scheme if for any $x \in X$, there is always some ring R and some open subset $x \in U \subseteq \text{Spec } R$ such that

23:4 Formalising Proj Construction in Lean

$(U, \mathcal{O}_X|_U) \cong (\text{Spec } R, \mathcal{O}_{\text{Spec } R})$ as locally ringed spaces. The category of schemes is the full subcategory of locally ringed spaces.

Thus to construct a scheme, one needs the following:

- a topological space X ;
- a presheaf \mathcal{O} ;
- a proof that \mathcal{O} satisfies the sheaf condition;
- a proof that all stalks are local;
- an open covering $\{U_i\}$ of X ;
- a collection of rings $\{R_i\}$ and isomorphism $(U_i, \mathcal{O}_X|_{U_i}) \cong (\text{Spec } R_i, \mathcal{O}_{\text{Spec } R_i})$.

In Section 2.3, Proj construction will be described following the steps above.

2.3 Proj Construction

Throughout this section, R will denote a ring and A an \mathbb{N} -graded R -algebra, in order to keep notations the same as Section 3, the grading of A will be written as \mathcal{A} , i.e. $A \cong \bigoplus_{i \in \mathbb{N}} \mathcal{A}_i$ as R -algebras.

Topology

► **Definition 5** (Proj \mathcal{A} as a set). Proj \mathcal{A} is defined to be $\{\mathfrak{p} \in \text{Spec } A \mid \mathfrak{p} \text{ is homogeneous and relevant}\}$, where

- an ideal $\mathfrak{p} \subseteq A$ is said to be homogeneous if for any $a \in \mathfrak{p}$ and $i \in \mathbb{N}$, a_i is in \mathfrak{p} as well where $a_i \in \mathcal{A}_i$ is the i -th projection of a with respect to grading \mathcal{A} ;
- an ideal $\mathfrak{p} \subseteq A$ is said to be relevant if $\bigoplus_{i=1}^{\infty} \mathcal{A}_i \not\subseteq \mathfrak{p}$.

Similar to Spec construction in Section 2.2, there is a topology on Proj \mathcal{A} whose closed sets are exactly the zero loci where for any $s \subseteq A$, zero locus of s is $\{\mathfrak{p} \in \text{Proj } \mathcal{A} \mid s \subseteq \mathfrak{p}\}$; this topology is also called the Zariski topology. For any $a \in A$, $D(a)$ denotes the set $\{x \in \text{Proj } \mathcal{A} \mid a \notin x\}$.

► **Theorem 6.** For any $a \in A$, $D(a)$ is open in Zariski topology and $\{D(a) \mid a \in A\}$ forms a basis of the Zariski topology [14].

Structure sheaf

Let $U \subseteq \text{Proj } \mathcal{A}$ be an open subset, the sections on U are defined to be

$$\mathcal{O}(U) = \left\{ s \in \prod_{x \in U} A_x^0 \mid s \text{ is locally a homogeneous fraction} \right\},$$

where $A_{\mathfrak{p}}^0$ denotes the homogeneous localization of A at a homogeneous prime ideal \mathfrak{p} , i.e. the subring of $A_{\mathfrak{p}}$ of elements of degree zero and s is said to be locally a homogeneous fraction if for any $x \in U$, there is some open subset $V \subseteq U$, $i \in \mathbb{N}$ and $a, b \in \mathcal{A}_i$ such that for all $y \in V$, $s(y) = \frac{a}{b}$. Equipped with natural restriction maps, \mathcal{O} defined in this way forms a presheaf; sheaf condition of \mathcal{O} is checked in the category of sets where it follows from that being locally a homogeneous fraction is a local predicate and local predicates define subsheaves. This sheaf is called the structure sheaf of Proj \mathcal{A} , also written as $\mathcal{O}_{\text{Proj } \mathcal{A}}$

Locally ringed space

► **Theorem 7.** *The stalk of $(\text{Proj } \mathcal{A}, \mathcal{O})$ at a homogeneous prime relevant ideal \mathfrak{p} is isomorphic to $A_{\mathfrak{p}}^0$.*

Proof. Let $U \ni \mathfrak{p}$ be an open subset of $\text{Proj } \mathcal{A}$, then a ring homomorphism $\mathcal{O}(U) \rightarrow A_{\mathfrak{p}}^0$ can be defined by evaluation at \mathfrak{p} , i.e. Since $\mathcal{O}_{\text{Proj } \mathcal{A}, \mathfrak{p}} = \text{colim}_{\mathfrak{p} \in U} \mathcal{O}(U)$, a ring homomorphism $f : \mathcal{O}_{\text{Proj } \mathcal{A}, \mathfrak{p}} \rightarrow A_{\mathfrak{p}}^0$ is obtained by universal property of colimit. To check that f is an isomorphism, it is sufficient to check bijectivity:

- Let $z_1 = \langle s_1, U_1 \rangle, z_2 = \langle s_2, U_2 \rangle \in \mathcal{O}_{\text{Proj } \mathcal{A}, \mathfrak{p}}$ be such that $f(z_1) = f(z_2) \stackrel{\text{def}}{\iff} s_1(\mathfrak{p}) = s_2(\mathfrak{p})$, then by definition of structure sheaf, there is some open subset $V \subseteq U_1 \cap U_2$ such that s_1 and s_2 are both constant on V . Since s_1, s_2 restrict to the same section on V , $z_1 = z_2$ hence proving injectivity.
- There is a function $A_{\mathfrak{p}}^0 \rightarrow \mathcal{O}_{\text{Proj } \mathcal{A}, \mathfrak{p}}$ defined by $\frac{a}{b} \mapsto \langle D(b), x \mapsto \frac{a}{b} \rangle$, this function is in fact a right inverse to f .

◀

Since $A_{\mathfrak{p}}^0$ is a local ring for any homogeneous prime ideal \mathfrak{p} , it can be concluded that $(\text{Proj } \mathcal{A}, \mathcal{O}_{\text{Proj } \mathcal{A}})$ is a locally ringed space.

Affine cover

► **Lemma 8.** *For any $x \in \text{Proj } \mathcal{A}$, there is some $0 < m \in \mathbb{N}$ and $f \in \mathcal{A}_m$, such that $x \in D(f) \stackrel{\text{def}}{\iff} f \notin x$.*

Proof. Let $x \in \text{Proj } \mathcal{A}$, by construction, $\bigoplus_{i=1}^{\infty} \mathcal{A}_i \not\subseteq x$. Thus there is some $f = f_1 + f_2 + \dots \notin x$, then at least one $f_i \notin x$ for otherwise $f \in x$.

◀

Thus, to construct an affine cover, it is sufficient to prove that for all $0 < m \in \mathbb{N}$ and homogeneous element $f \in \mathcal{A}_m$, $(D(f), \mathcal{O}_{\text{Proj } \mathcal{A}}|_{D(f)}) \cong (\text{Spec } A_f^0, \mathcal{O}_{\text{Spec } A_f^0})$ where A_f^0 is the subring of the localised ring A_f consisted of elements of degree zero. Now fix these notations, an isomorphism between locally ringed space is a pair (ϕ, α) where ϕ is a homeomorphism between topological space $D(f) \cong \text{Spec } A_f^0$ and α an isomorphism between $\phi_*(\mathcal{O}_{\text{Proj } \mathcal{A}}|_{D(f)}) \cong \mathcal{O}_{\text{Spec } A_f^0}$.

► **Theorem 9.** *$D(f) \cong \text{Spec } A_f^0$ are homeomorphic as topological spaces.*

The following proofs are an expansion of [9, II.2.5] while drawing ideas from [15, II.4.5].

Proof. Define $\phi : D(f) \rightarrow \text{Spec } A_f^0$ by $\mathfrak{p} \mapsto \text{span} \left\{ \frac{g}{1} \mid g \in \mathfrak{p} \right\} \cap A_f^0$; by clearing denominators, one can show that $\phi(\mathfrak{p}) = \text{span} \left\{ \frac{g}{f^i} \mid g \in \mathfrak{p} \cap \mathcal{A}_{mi} \right\}$. One can check that $\phi(\mathfrak{p})$ is indeed a prime ideal. ϕ is continuous by checking on the topological basis consisting of basic open sets of $\text{Spec } A_f^0$. The fact that basic open sets form a basis is already recorded in mathlib [5]. Take $\frac{a}{f^n} \in A_f^0$, then $\phi^{-1}(D(a/f^n)) = D(f) \cap D(a)$.

- $D(f) \cap D(a) \subseteq \phi^{-1}(D(a/f^n))$ because if $y \in D(f) \cap D(a)$ and $a/f^n \in \phi(y)$, i.e. $a/f^n = \sum_i c_i / f^{n_i} g_i / 1$, then by multiplying suitable powers of f , $a f^N / 1 = \sum_i c_i g_i f^{m_i} / 1$ for some N , by definition of localisation, $a f^N f^M = \sum_i c_i g_i f^{m_i}$ for some M implying that $a \in y$. Contradiction.
- On the other hand, if $\phi(y) \in D(a/f^n)$ and $a \in y$, then $a/1 \in \phi(y)$, contradiction because $a/f^n = a/1/f^n \in \phi(y)$.

For the other direction, define $\psi : \text{Spec } A_f^0 \rightarrow D(f)$ to be $x \mapsto \left\{ a \mid \text{for all } i \in \mathbb{N}, \frac{a_i^m}{f^i} \in x \right\}$. For ψ to be well-defined, one needs to check that $\psi(x)$ is a homogeneous prime ideal that is relevant. Continuity of ψ depends on that ϕ and ψ are inverse to each other. $D(f)$ with the subspace topology has a basis of the form $D(f) \cap D(a)$, thus it is sufficient to prove that preimages of these sets are open. By considering $\phi(D(f) \cap D(a)) = \bigcup_i \phi(D(f) \cap D(a_i))$, each $\phi(D(f) \cap D(a_i))$ is open because $\phi(D(f) \cap D(a_i)) = D(a_i^m/f^i)$ in $\text{Spec } A_f^0$. To prove $\phi(D(f) \cap D(a_i)) = D(a_i^m/f^i)$, it is sufficient to prove $\phi^{-1}(D(a_i^m/f^i)) = D(f) \cap D(a)$ and this is proven in continuity of ϕ . Since ϕ and ψ are inverses to each other, preimage of $D(f) \cap D(a)$ is indeed $\phi(D(f) \cap D(a))$. ◀

Let ϕ and ψ be the continuous functions defined in the previous proof, U be an open subset of $\text{Spec } A_f^0$, s be a section on $\phi^{-1}(U)$ and $x \in U$, then $\psi(x) \in \phi^{-1}(U)$, hence $s(\psi(x)) = \frac{n}{d} \in A_{\psi(x)}^0$ for some $i \in \mathbb{N}$ and $n, d \in \mathcal{A}_i$. Keeping the same notation, a ring homomorphism $\alpha_U : \phi_*(\mathcal{O}_{\text{Proj } \mathcal{A}}|_{D(f)})(U) \rightarrow \mathcal{O}_{\text{Spec } A_f^0}(U)$ can be defined as $s \mapsto \left(x \mapsto \frac{nd^{m-1}/f^i}{d^m/f^i} \right)$ where $n, d \in \mathcal{A}_i$. Assuming α_U is well-defined, it is easy to check that $U \mapsto \alpha_U$ is natural in U , hence α defines a morphism of sheaves³.

► **Lemma 10.** *For any open subset $U \subseteq \text{Spec } A_f^0$, α_U is well-defined; hence α defines a morphism of sheaves.*

Proof. It is clear that both the numerator and denominator have degrees zero. $d^m/f^i \notin x$ follows from $d \notin \psi(x)$. $\alpha_U(s)$ is locally a fraction: since s is locally a quotient, for any $x \in U$, there is some open set $V \subseteq \text{Proj } \mathcal{A}$ such that $\psi(x) \in V \subseteq \phi^{-1}(U)$ such that $s(y) = \frac{a}{b}$ for all $y \in V$ where $a, b \in A_n$ and $b \notin y$, then to check $\alpha_U(s)$ is locally quotient, use the open subset $\phi(V)$ and check that for all $z \in \phi(V)$, $\alpha_U(s)(z) = \frac{ab^{m-1}}{b^m}$. The proof of α_U being a ring homomorphism involves manipulations of fractions in localised rings, for more details, see Section 3. ◀

In the other direction, if $s \in \mathcal{O}_{\text{Spec } A_f^0}(U)$ and $y \in \phi^{-1}(U)$, then $\phi(y) \in U$, so $s(\phi(y))$ can be written as $\frac{a}{b}$ where $a, b \in A_f^0$; then a can be written as $\frac{n_a}{f^{i_a}}$ for some $n_a \in A_{mi_a}$ and b as $\frac{n_b}{f^{i_b}}$ for some $n_b \in A_{mi_b}$. Hence, a ring homomorphism $\beta_U : \mathcal{O}_{\text{Spec } A_f^0}(U) \rightarrow \mathcal{O}_{\text{Proj } \mathcal{A}}(\phi^{-1}(U))$ can be defined as $s \mapsto \left(y \mapsto \frac{n_a f_b^{i_a}}{n_b f_a^{i_b}} \right)$. Assuming β is well defined, it is easy to check that the assignment $U \mapsto \beta_U$ is natural so that β is a natural transformation⁴.

► **Lemma 11.** *For any open subset $U \subseteq \text{Spec } A_f^0$, β_U is well-defined; hence β defines a morphism of sheaves.*

Proof. $n_a f_b^{i_a}$ and $n_b f_a^{i_b}$ have the same degree. $n_b f_a^{i_b} \notin y$ follows from $b \notin \phi(y)$. Since s locally is a fraction, there are open sets $\phi(y) \in V \subseteq U$, such that for all $z \in V$, $s(z)$ is $\frac{a/f^{l_1}}{b/f^{l_2}}$. Then on $\phi^{-1}(V) \subseteq \phi^{-1}(U)$, $\psi_U(s)(y)$ is always $\frac{a f^{l_2}}{b f^{l_1}}$. Checking that β_U is a ring homomorphism involves manipulating fractions of fractions. ◀

► **Theorem 12.** *$\phi_*(\mathcal{O}_{\text{Proj } \mathcal{A}}|_{D(f)})$ and $\mathcal{O}_{\text{Spec } A_f^0}$ are isomorphic as sheaves.*

Proof. By combining Lemma 10 and Lemma 11, it is sufficient to check $\alpha \circ \beta$ and $\beta \circ \alpha$ are both identities.

³ In fact, `simp` can do this.

⁴ In fact, `simp` can do this.

- $\beta \circ \alpha = 1$: let $s \in \mathcal{O}_{\text{Proj } \mathcal{A}}|_{D(f)}(\phi^{-1}(U))$, then for $x \in \phi^{-1}(U)$

$$\alpha_U(s) = x \mapsto \frac{nd^{m-1}/f^i}{d^m/f^i},$$

where $s(x) = \frac{n}{d}$. Thus, by definition

$$\beta_U(\alpha_U(s))(x) = \frac{nd^{m-1}f^i}{d^m f^i} = \frac{n}{d} = s(x).$$

- $\alpha \circ \beta = 1$: let $s \in \mathcal{O}_{\text{Spec } A_f^0}(U)$, then for $x \in U$

$$\beta_U(s) = x \mapsto \frac{n_a f^{i_b}}{n_b f^{i_a}}$$

where $s(x) = \frac{n_a/f^{i_a}}{n_b/f^{i_b}}$. Thus

$$\phi_U(\psi_U(s))(x) = \frac{n_a f^{i_b} (n_b f^{i_a})^{m-1} / f^j}{(n_b f^{i_a})^m / f^j} = \frac{n_a / f^{i_a}}{n_b / f^{i_b}} = s(x).$$

◀

► **Corollary 13.** $(\text{Proj } \mathcal{A}, \mathcal{O}_{\text{Proj } \mathcal{A}})$ is a scheme.

3 Formalisation details

3.1 Homogeneous Ideal

Let A be an R -algebra and an ι -grading $\mathcal{A} : \iota \rightarrow R$ -submodules of A [17], `ideal.is_homogeneous` is the proposition of an ideal being homogeneous and `homogeneous_ideal` is the type of all homogeneous ideals of A . Note that, by this implementation, homogeneous ideals are not literally ideals, for this reason, $\text{Proj } \mathcal{A}$ cannot be implemented as a subset of $\text{Spec } A$.

```
def ideal.is_homogeneous : Prop :=
  ∀ (i : ι) {r : A}, r ∈ I → (direct_sum.decompose A r i : A) ∈ I

structure homogeneous_ideal extends submodule A A :=
  (is_homogeneous' : ideal.is_homogeneous A to_submodule)

def homogeneous_ideal.to_ideal (I : homogeneous_ideal A) : ideal A :=
  I.to_submodule

lemma homogeneous_ideal.is_homogeneous (I : homogeneous_ideal A) :
  I.to_ideal.is_homogeneous A := I.is_homogeneous'

def homogeneous_ideal.irrelevant : homogeneous_ideal A :=
  ⟨(graded_ring.proj_zero_ring_hom A).ker, sorry⟩
```

3.2 Homogeneous Localisation

If x is a multiplicatively closed subset, then the homogeneous localisation of A at x is defined to be the subring of localised ring A_x consisting of elements of degree zero. This ring is implemented as triples $\{(i, a, b) : \iota \times \mathcal{A}_i \times \mathcal{A}_i \mid b \notin x\}$ under the equivalence relation that

$(i_1, a_1, b_1) \approx (i_2, a_2, b_2) \stackrel{\text{def}}{\iff} \frac{a_1}{b_1} = \frac{a_2}{b_2} \text{ in } A_x$. This approach gives an induction principle via quotients, though the construction still uses classical reasoning, many lemmas will be automatic because of rich APIs in `mathlib` about quotient spaces already; compared to the subring approach, one would need to write corresponding lemmas manually by excessively invoking `classical.some` and `classical.some_spec`. One potential benefit of the subring approach is that different propositions can be specified for different multiplicative subsets to customize what properties and attributes are to be made explicit; for example for localisation away from a single element, it is useful to make powers of denominator explicit. But this would sacrifice a universal approach to homogeneous localisation for different multiplicative subsets so that auxiliary lemmas would have to be duplicated. To maintain consistency and prevent duplication, this paper will adopt the approach from quotient space. Before writing this paper, the subring approach has also been tested, by comparing the two approaches, it proves that there is no significant difference in the smoothness of two formalisations but the quotienting approach has smaller code sizes.

```
variables {ι R A: Type*} [add_comm_monoid ι] [decidable_eq ι]
variables [comm_ring R] [comm_ring A] [algebra R A]
variables (A : ι → submodule R A) [graded_algebra A]
variables (x : submonoid A)

structure num_denom_same_deg :=
  (deg : ι) (num denom : A deg) (denom_mem : (denom : A) ∈ x)

def embedding (p : num_denom_same_deg A x) : localization x :=
  localization.mk p.num ⟨p.denom, p.denom_mem⟩

def homogeneous_localization : Type* := quotient (setoid.ker $ embedding A x)
```

Then if $(y : \text{homogeneous_localization } A \ x)$, its value, degree, numerator and denominator can all be defined by using induction/recursion principles for quotient spaces:

```
variable (y : homogeneous_localization A x)

def val : localization x :=
  quotient.lift_on' y (num_denom_same_deg.embedding A x) $ λ _ _, id

def num : A := (quotient.out' y).num
def denom : A := (quotient.out' y).denom
def deg : ι := (quotient.out' y).deg

lemma denom_mem : y.denom ∈ x := (quotient.out' y).denom_mem
lemma num_mem_deg : y.num ∈ A f.deg := (quotient.out' y).num.2
lemma denom_mem_deg : y.denom ∈ A y.deg := (quotient.out' y).denom.2
lemma eq_num_div_denom : y.val = localization.mk y.num ⟨y.denom, y.denom_mem⟩ :=
  sorry
```

3.3 Zariski Topology

In this section A will be graded by \mathbb{N} and the grading denoted by \mathcal{A} . $\text{Proj } \mathcal{A}$ is formalised a structure:

```
structure projective_spectrum :=
  (as_homogeneous_ideal : homogeneous_ideal A)
```



```
(is_prime : as_homogeneous_ideal.to_ideal.is_prime)
(not_irrelevant_le : ¬(homogeneous_ideal.irrelevant  $\mathcal{A} \leq$  as_homogeneous_ideal))
```

After building more APIs around `projective_spectrum`, Zariski topology with a basis of basic open sets can be formalised as:

```
def zero_locus (s : set A) : set (projective_spectrum A) :=
{x | s  $\subseteq$  x.as_homogeneous_ideal}

instance zariski_topology : topological_space (projective_spectrum A) :=
topological_space.of_closed (set.range (zero_locus A)) sorry sorry sorry

def basic_open (r : A) : topological_space.opens (projective_spectrum A) :=
{ val := { x | r  $\notin$  x.as_homogeneous_ideal },
  property := ⟨{r}, set.ext $  $\lambda$  x, set.singleton_subset_iff.trans $ not_not.symm⟩ }

lemma is_topological_basis_basic_opens : topological_space.is_topological_basis
(set.range ( $\lambda$  (r : A), (basic_open A r : set (projective_spectrum A)))) :=
sorry
```

3.4 Locally Ringed Space

`mathlib` provides `Top.presheaf.is_sheaf_iff_is_sheaf_comp` to check sheaf condition by composing a forgetful functor and `Top.subsheaf_to_Types` to construct subsheaf of types satisfying a local predicate [6]; $\mathcal{O}_{\text{Spec}}$ in `mathlib` adopted this approach [5], and structure sheaf of `Proj` will also be constructed in this way. `is_locally_fraction` is a local predicate expressing “being locally a homogeneous fraction” in Section 2.3:

```
def is_fraction {U : opens (Proj A)} (f :  $\prod$  x : U,  $A_x^0$ ) : Prop :=
 $\exists$  (i :  $\mathbb{N}$ ) (r s : A i),  $\forall$  x : U,  $\exists$  (s_nin : s.1  $\notin$  x.1.as_homogeneous_ideal),
  f x = quotient.mk' ⟨i, r, s, s_nin⟩

def is_fraction_prelocal : prelocal_predicate ( $\lambda$  (x : Proj A),  $A_x^0$ ) :=
{ pred :=  $\lambda$  U f, is_fraction f,
  res := by rintros V U i f ⟨j, r, s, w⟩; exact ⟨j, r, s,  $\lambda$  y, w (i y)⟩ }

def is_locally_fraction : local_predicate ( $\lambda$  (x : Proj A),  $A_x^0$ ) :=
(is_fraction_prelocal A).sheafify

def structure_sheaf_in_Type : sheaf Type* (Proj A) :=
subsheaf_to_Types (is_locally_fraction A)
```

The presheaf of rings is also defined as `structure_presheaf_in_CommRing` and checked that composition with forgetful functor is naturally isomorphic to the underlying presheaf of `structure_sheaf_in_Type` which implies that `structure_presheaf_in_CommRing` satisfies the sheaf condition as well by using `Top.presheaf.is_sheaf_iff_is_sheaf_comp`.

```
def structure_presheaf_in_CommRing : presheaf CommRing (Proj A) :=
{ obj :=  $\lambda$  U, CommRing.of ((structure_sheaf_in_Type A).1.obj U), ..sorry }

def structure_presheaf_comp_forget :
  structure_presheaf_in_CommRing A  $\ggg$  (forget CommRing)  $\cong$ 
  (structure_sheaf_in_Type A).1 :=
sorry
```

```
def Proj.structure_sheaf : sheaf CommRing (Proj A) :=
  ⟨structure_presheaf_in_CommRing A, (is_sheaf_iff_is_sheaf_comp _ _).mpr
    (is_sheaf_of_iso (structure_presheaf_comp_forget A).symm
      (structure_sheaf_in_Type A).cond)⟩
```

Then following Theorem 7, `stalk_to_fiber_ring_hom` is a family of ring homomorphism $\prod_x \mathcal{O}_{\text{Proj } A, x} \rightarrow A_x^0$ obtained by universal property of colimit with its right inverse as a family of function `homogeneous_localization_to_stalk`:

```
def stalk_to_fiber_ring_hom (x : Proj A) :
  (Proj.structure_sheaf A).presheaf.stalk x → CommRing.of A_x^0 :=
  limits.colimit.desc (((open_nhds.inclusion x).op) >>> (Proj.structure_sheaf A).1)
  sorry

def section_in_basic_open (x : Proj A) :
  Π (f : A_x^0), (Proj.structure_sheaf A).1.obj (op (Proj.basic_open A f.denom)) :=
  λ f, ⟨λ y, quotient.mk' ⟨_, ⟨f.num, _⟩, ⟨f.denom, _⟩, _⟩, _⟩

def homogeneous_localization_to_stalk (x : Proj A) :
  A_x^0 → (Proj.structure_sheaf A).presheaf.stalk x :=
  λ f, (Proj.structure_sheaf A).presheaf.germ
    (⟨x, homogeneous_localization.mem_basic_open _ x f⟩ : Proj.basic_open _ f.denom)
    (section_in_basic_open _ x f)

def Proj.stalk_iso' (x : Proj A) :
  (Proj.structure_sheaf A).presheaf.stalk x ≃+* CommRing.of A_x^0 :=
  ring_equiv.of_bijective (stalk_to_fiber_ring_hom _ x)
  ⟨sorry, function.surjective_iff_has_right_inverse.mpr
    (homogeneous_localization_to_stalk A x, sorry)⟩
```

Hence establishing that `Proj A` is a locally ringed space:

```
def Proj.to_LocallyRingedSpace : LocallyRingedSpace :=
  { local_ring := λ x, @@ring_equiv.local_ring _
    (show local_ring A_x^0, from infer_instance) _
    (Proj.stalk_iso' A x).symm,
    ..(Proj.to_SheafedSpace A) }
```

3.5 Affine cover

```
variables {f : A} {m : ℕ} (f_deg : f ∈ A m) (x : Proj | D(f))
```

`Spec.T` and `Proj.T` denotes the topological space associated with each locally ringed spaces. Let $0 < m \in \mathbb{N}$ and $f \in \mathcal{A}_m$ and $x \in D(f)$, by following Theorem 9, the continuous function ϕ and ψ in Section 2.3 is formalised as `Proj_iso_Spec_Top_component.to_Spec` and `Proj_iso_Spec_Top_component.from_Spec` respectively; $\phi \circ \psi = 1$ and $\psi \circ \phi = 1$ are recorded as `Proj_iso_Spec_Top_component.to_Spec_from_Spec` and `.from_Spec_to_Spec` respectively:

```
namespace Proj_iso_Spec_Top_component
namespace to_Spec

def carrier : ideal A_f^0 :=
  ideal.comap (algebra_map A_f^0 A_f)
```

```

    (ideal.span $ algebra_map A (away f) '' x.val.as_homogeneous_ideal)

def to_fun : Proj.T | D(f) → Spec.T  $A_f^0$  :=
λ x, ⟨carrier  $\mathcal{A}$  x, sorry /-a proof for primeness-/⟩

end to_Spec

def to_Spec (f : A) : Proj.T | D(f) → Spec.T  $A_f$  :=
{ to_fun := to_Spec.to_fun  $\mathcal{A}$  f,
  continuous_to_fun := begin
    apply is_topological_basis.continuous (Spec.is_topological_basis_basic_opens);
    sorry
  end }

namespace from_Spec

def carrier (q : Spec.T  $A_f^0$ ) : set A :=
{a | ∀ i, (quotient.mk' ⟨_, ⟨proj  $\mathcal{A}$  i a ^ m, _⟩, ⟨f^i, _⟩, _⟩ :  $A_f^0$ ) ∈ q.1}

def carrier.as_ideal : ideal A := { carrier := carrier f_deg q, ..sorry }
def carrier.as_homogeneous_ideal : homogeneous_ideal  $\mathcal{A}$  :=
⟨carrier.as_ideal f_deg hm q, sorry⟩

lemma carrier.relevant :
  -homogeneous_ideal.irrelevant  $\mathcal{A} \leq$  carrier.as_homogeneous_ideal f_deg hm q :=
sorry

lemma carrier.as_ideal.prime : (carrier.as_ideal f_deg hm q).is_prime :=
(carrier.as_ideal.homogeneous f_deg hm q).is_prime_of_homogeneous_mem_or_mem
  sorry sorry

def to_fun : Spec.T  $A_f^0$  → Proj.T | D(f) :=
λ q, ⟨⟨carrier.as_homogeneous_ideal f_deg hm q, carrier.as_ideal.prime f_deg hm q,
  carrier.relevant f_deg hm q⟩, sorry⟩

end from_Spec

lemma to_Spec_from_Spec : to_Spec.to_fun  $\mathcal{A}$  f (from_Spec.to_fun f_deg hm x) = x :=
sorry
lemma from_Spec_to_Spec : from_Spec.to_fun f_deg hm (to_Spec.to_fun  $\mathcal{A}$  f x) = x :=
sorry

def from_Spec : Spec.T  $A_f^0$  → Proj.T | D(f) :=
{ to_fun := from_Spec.to_fun f_deg hm,
  continuous_to_fun := begin
    apply is_topological_basis.continuous,
    sorry
  end }

end Proj_iso_Spec_Top_component

def Proj_iso_Spec_Top_component :
  Proj.T | D(f)  $\cong$  Spec.T ( $A_f^0$ ) :=
{ hom := Proj_iso_Spec_Top_component.to_Spec  $\mathcal{A}$  f,
  inv := Proj_iso_Spec_Top_component.from_Spec hm f_deg,

```

23:12 Formalising Proj Construction in Lean

```
..sorry /-composition being identity-/ }
```

Then by following Lemma 11, β is formalised as `Proj_iso_Spec_Sheaf_component.from_Spec`:

```
namespace Proj_iso_Spec_Sheaf_component
namespace from_Spec

variables (V : (opens (Spec A_f^0))^op) (s : (Spec A_f^0).presheaf.obj V)
variables (y : ((@opens.open_embedding Proj.T D(f)).is_open_map.functor.op.obj
  ((opens.map (Proj_iso_Spec_Top_component hm f_deg).hom).op.obj V)).unop)
-- This is but a verbose way of spelling y is in  $\phi^{-1}(V)$  for type checking reasons.

-- Corresponding to evaluating a section in Lemma 11.  $s(\phi(y))$ 
def data : structure_sheaf.localizations A_f^0
  ((Proj_iso_Spec_Top_component hm f_deg).hom ⟨y.1, _⟩) :=
s.1 ⟨_, _⟩

--  $s(\phi(y)) = \frac{a}{b}$ , this is  $a$ , see Lemma 11.
def data.num : A_f^0 := sorry

--  $s(\phi(y)) = \frac{a}{b}$ , this is  $b$ , see Lemma 11
def data.denom : A_f^0 := sorry

--  $s \mapsto (y \mapsto \frac{n_a f_b^i}{n_b f_b^{ia}})$ , this is  $n_a f_b^i$ , see Lemma 11.
def num : A :=
  (data.num _ hm f_deg s y).num * (data.denom _ hm f_deg s y).denom

--  $s \mapsto (y \mapsto \frac{n_a f_b^i}{n_b f_b^{ia}})$ , this is  $n_b f_b^{ia}$ , see Lemma 11.
def denom : A :=
  (data.denom _ hm f_deg s y).num * (data.num _ hm f_deg s y).denom

--  $s \mapsto (y \mapsto \frac{n_a f_b^i}{n_b f_b^{ia}})$ , this is  $\frac{n_a f_b^i}{n_b f_b^{ia}}$ , see Lemma 11.
def bmk : A_y^0 :=
quotient.mk'
{ deg := (data.num _ hm f_deg s y).deg + (data.denom _ hm f_deg s y).deg,
  num := ⟨num hm f_deg s y, _⟩,
  denom := ⟨denom hm f_deg s y, _⟩,
  denom_mem := sorry }

def to_fun.aux : ((Proj_iso_Spec_Top_component hm f_deg).hom _ * (Proj|
  D(f)).presheaf).obj V :=
⟨bmk hm f_deg V s, sorry /-being locally a homogeneous fraction-/⟩

def to_fun : (Spec A_f^0).presheaf.obj V →
  ((Proj_iso_Spec_Top_component hm f_deg).hom _ * (Proj| D(f)).presheaf).obj V :=
{ to_fun := λ s, to_fun.aux A hm f_deg V s, ..sorry /-ring homomorphism proofs-/ }

end from_Spec

def from_Spec : (Spec A_f^0).presheaf →
  (Proj_iso_Spec_Top_component hm f_deg).hom _ * (Proj| D(f)).presheaf :=
{ app := λ V, from_Spec.to_fun A hm f_deg V,
  naturality' := λ _ _ _, by { ext1, simpa } }

end Proj_iso_Spec_Sheaf_component
```

By following Lemma 10, α is formalised as `Proj_iso_Spec_Sheaf_component.to_Spec`:

```

namespace Proj_iso_Spec_Sheaf_component
namespace to_Spec
variable (U : (opens (Spec.T A_f^0))^op)
variable (s : ((Proj_iso_Spec_Top_component hm f_deg).hom _*
  (Proj| D(f))).presheaf.obj U) -- (phi_*(O_Proj|_{D(f)}))(U)

-- evaluating a section, this is s(psi(y))
def hl (y : unop U) : homogeneous_localization A _ :=
s.1 <((Proj_iso_Spec_Top_component hm f_deg).inv y.1).1, _>

-- s maps (x maps to n d^{m-1}/f^i/d^m/f^i) where n,d in A_i, this is n d^{m-1}/f^i, see Lemma 10.
def num (y : unop U) : A_f^0 :=
quotient.mk'
{ deg := m * (hl hm f_deg s y).deg,
  num := <(hl hm f_deg s y).num * (hl hm f_deg s y).denom ^ m.pred, _>,
  denom := <f^(hl hm f_deg s y).deg, _>,
  denom_mem := _ }

-- s maps (x maps to n d^{m-1}/f^i/d^m/f^i) where n,d in A_i, this is d^m/f^i, see Lemma 10.
def denom (y : unop U) : A_f^0 :=
quotient.mk'
{ deg := m * (hl hm f_deg s y).deg,
  num := <(hl hm f_deg s y).denom ^ m, _>,
  denom := <f ^ (hl hm f_deg s y).deg, _>,
  denom_mem := _ }

-- s maps (x maps to n d^{m-1}/f^i/d^m/f^i) where n,d in A_i, this is n d^{m-1}/f^i/d^m/f^i, see Lemma 10.
def fmk (y : unop U) : (A_f^0)_y :=
mk (num hm f_deg s y) <denom hm f_deg s y, _>

def to_fun :
  ((Proj_iso_Spec_Top_component hm f_deg).hom _* (Proj| D(f))).obj U ->
  (Spec A_f^0).presheaf.obj U :=
{ to_fun := lambda s, <lambda y, fmk hm f_deg s y, sorry /-proof of being locally a
  fraction-/>, ..sorry /-proof of being a ring homomorphism-/>,
end to_Spec

def to_Spec :
  (Proj_iso_Spec_Top_component hm f_deg).hom _* (Proj| D(f)).presheaf ->
  (Spec A_f^0).presheaf :=
{ app := lambda U, to_Spec.to_fun hm f_deg U,
  naturality' := lambda U V subset1, by { ext1, simpa } }
end Proj_iso_Spec_Sheaf_component

```

Hence it has been shown that $(D(f), \mathcal{O}_{\text{Proj } \mathcal{A}}) \cong (\text{Spec } A_f^0, \mathcal{O}_{\text{Spec } A_f^0})$ as locally ringed spaces and hence $(\text{Proj } \mathcal{A}, \mathcal{O}_{\text{Proj } \mathcal{A}})$ is a scheme.

```

def Sheaf_component :
  (Proj_iso_Spec_Top_component hm f_deg).hom _* (Proj| D(f)).presheaf ≅
  (Spec A_f^0).presheaf :=
{ hom := Proj_iso_Spec_Sheaf_component.to_Spec A hm f_deg,
  inv := Proj_iso_Spec_Sheaf_component.from_Spec A hm f_deg,
  ..sorry /-composition is identity-/ }

```

```

def iso :
  (Proj| D(f)) ≅ Spec A_f^0 :=
let H : (Proj| D(f)).to_PresheafedSpace ≅ (Spec A_f^0).to_PresheafedSpace :=
  PresheafedSpace.iso_of_components
  (Proj_iso_Spec_Top_component hm f_deg) (Sheaf_component A f_deg hm) in
LocallyRingedSpace.iso_of_SheafedSpace_iso
{ hom := H.1, inv := H.2, hom_inv_id' := H.3, inv_hom_id' := H.4 }

def Proj.to_Scheme : Scheme :=
{ local_affine := sorry, ..Proj }

```

3.6 Reflection

An example of calculation

Most calculations in proofs of Theorem 9 and Lemmas 10 and 11 are omitted. I present the details of verifying β_U preserves multiplication to showcase the flavour of calculations involved. Verifying that β_U preserving zero and one is similar but slightly simpler while preservation of addition is more cumbersome. Since α only involves one layer of fractions, calculations are not as long.

Let x, y be two sections, the aim is to show $\beta_U(xy) = \beta_U(x)\beta_U(y)$, i.e. for all $z \in \phi^{-1}(U)$, $\beta_U(xy)(z) = \beta_U(x)(z)\beta_U(y)(z)$.

```

lemma bmk_mul (x y : (Spec A_f^0).presheaf.obj V) :
  bmk hm f_deg V (x * y) = bmk hm f_deg V x * bmk hm f_deg V y :=
begin
  ext1 z,

```

by writing $x(\phi(z))$ as $\frac{a_x/f^{i_x}}{b_x/f^{j_x}}$, $y(\phi(z))$ as $\frac{a_y/f^{i_y}}{b_y/f^{j_y}}$ and $(xy)(\phi(z)) = \frac{a_{xy}/f^{i_{xy}}}{b_{xy}/f^{j_{xy}}}$, one deduces that $\frac{a_x a_y / f^{i_x + i_y}}{b_x b_y / f^{j_x + j_y}} = \frac{a_{xy} / f^{i_{xy}}}{b_{xy} / f^{j_{xy}}}$, by definition of equality in localised ring, it implies that, there is some $\frac{c}{f^l}$ such that

$$\frac{a_x a_y b_{xy} c}{f^{i_x + i_y + j_{xy} + l}} = \frac{a_{xy} b_x b_y c}{f^{i_{xy} + j_x + j_y + l}}.$$

```

have mul_eq := data.eq_num_div_denom hm f_deg (x * y) z,
... -- simplification
erw is_localization.eq at mul_eq,
obtain ⟨⟨C, hC⟩, mul_eq⟩ := mul_eq, -- C is the c/f^l above.
...

-- c ∉ z
have C_not_mem : C.num ∉ z.1.as_homogeneous_ideal := sorry,

-- setting up notations.
set a_xy := _, set i_xy := _, set b_xy := _, set j_xy := _,
set a_x := _, set i_x := _, set b_x := _, set j_x := _,
set a_y := _, set i_y := _, set b_y := _, set j_y := _,
set l := _,
...

```

By definition of equality in localisation again, there exists some $n_1 \in \mathbb{N}$ such that

$$a_x a_y b_{xy} c f^{i_x y + j_x + j_y + l + n_1} = a_{xy} b_x b_y c f^{i_{xy} + i_y + j_{xy} + l + n_1} \quad (1)$$

```
... -- more simplification
obtain ⟨⟨_, ⟨n1, rfl⟩⟩, mul_eq⟩ := mul_eq,
...
```

The aim is to show

$$\frac{a_{xy} f^{j_{xy}}}{b^{xy} f^{i_x}} = \frac{a_x f^{j_x}}{b^x f^{i_x}} \frac{a_y f^{j_y}}{b_y f^{i_y}},$$

by Equation (1) and definition of equality in localised ring, cf^{l+n_1} verifies this equality.

```
suffices : (mk (a_xy * f ^ j_xy) ⟨b_xy * f ^ i_xy, _⟩ : localization.at_prime _)
= mk (a_x * f ^ j_x) ⟨b_x * f ^ i_x, _⟩ * mk (a_y * f ^ j_y) ⟨b_y * f ^ i_y, _⟩
:= sorry,
...
refine ⟨⟨C.num * f^(1 + n1), _⟩, _⟩,
...
end
```

In totality, this is about ~100 lines of code by following essentially three lines of calculation when done with pen-and-papers. Admittedly, the above code is not the most optimal, but the magnitude is not greatly exaggerated. I think the following factors contribute to the differences between formalisation and a pen-and-paper-proof:

- Every element of a localised ring can be written as a fraction of a numerator and a denominator is a corollary of the construction but not straightly from its definition. Thus, many extra steps are required to set up the proof.
- Elements of a (homogeneously) localised ring contains not only data, but proofs as well. For example, the denominator of an element is a term $\langle d, \text{some_proof} \rangle$ of a subtype. This makes `rewrite` less smooth to use, for equalities are often of the form $h : d = d'$, thus `rewrite [h]` is type theoretically unsound.
- Terms of `localization x` or `homogeneous_localization A x` have to contain proofs to make the definitions correct, thus constructing any term of these types requires many proofs or disproofs of membership. Thus, a formalised calculation cannot be as liberal as a pen-and-paper-proof when comes whether the terms are well-defined.
- Not many high powered tactics are available for localised ring, for example `ring` will be able to solve $x * y = y * x$ and much more complicated goal in a commutative ring, but `ring` cannot (and should not be able to) solve $(a / b * c / d : \text{localization } _) = c / b * a / d$.

The first three bullet points are essentially all because formalisations require more rigour than that of pen-and-paper-proofs; whether the requirement of extra rigour is beneficial is another question and not in the scope of this paper. For the fourth bullet point, it is definitely helpful to have a tactic automating many proofs, the catch is that equality in localised ring is existentially quantified — $\frac{a}{b} = \frac{a'}{b'}$ if and only if $ab'c = a'bc$ for some c in a multiplicative subset. Thus, a tactic can only do so much without human input for now.

On propositional equality

Originally, I expected propositional equalities that are not equal by definition such as $\phi(\psi(y)) = y$ in Theorem 9 will pose a challenge, but the difficulty is less severe: indeed, I only need to prove some redundant lemma like $\phi(\psi(y))$ is in some open sets that clearly contains y ; the reason is that in this project I did not compare algebraic structures depending on propositional equality, i.e. \mathcal{O}_y and $\mathcal{O}_{\phi(\psi(y))}$; but foreseeably, this difficulty will come back when one starts to develop the theory of projective variety furtherer.

4 Conclusion

Since a large part of modern algebraic geometry depends on Proj construction, much potential future research is possible: calculating cohomology of projective spaces; defining projective morphisms; Serre’s twisting sheaves to name a few. Other approaches to Proj construction also exists, for example, by gluing a family of schemes together; however, since there is no other formalisation of Proj construction, I could not compare different approaches or compare capabilities of formalising modern algebraic geometry of different theorem provers. Thus I would like to conclude this paper with an invitation/challenge — state and formalise something involving more than affine schemes in your preferred theorem prover; for the only way to know which, if any, theorem provers handles modern mathematics satisfactorily is to actually formalise more modern mathematics.

References

- 1 Anthony Bordg, Lawrence Paulson, and Wenda Li. Simple type theory is not too simple: Grothendieck’s schemes without dependent types. *Experimental Mathematics*, 31(2):364–382, 2022. arXiv:<https://doi.org/10.1080/10586458.2022.2062073>, doi:10.1080/10586458.2022.2062073.
- 2 Kevin Buzzard, Chris Hughes, Kenny Lau, Amelia Livingston, Ramon Fernández Mir, and Scott Morrison. Schemes in lean. *Experimental Mathematics*, 31(2):355–363, 2022.
- 3 Laurent Chicli. *Une formalisation des faisceaux et des schémas affines en théorie des types avec Coq*. PhD thesis, INRIA, 2001.
- 4 Wei-Liang Chow. On compact complex analytic varieties. *American Journal of Mathematics*, 71(4):893–914, 1949. URL: <http://www.jstor.org/stable/2372375>.
- 5 Mathlib Contributors. Lean mathlib. <https://github.com/leanprover-community/mathlib>, 2022.
- 6 Mathlib Contributors. Mathlib documentation. https://leanprover-community.github.io/mathlib_docs, 2023.
- 7 Leonardo de Moura, Soonho Kong, Jeremy Avigad, Floris Van Doorn, and Jakob von Raumer. The lean theorem prover (system description). In *International Conference on Automated Deduction*, pages 378–388. Springer, 2015.
- 8 Roger Godement. Topologie algébrique et théorie des faisceaux. *Publications de*, 1, 1958.
- 9 Robin Hartshorne. Graduate texts in mathematics. *Algebraic Geometry*, 52, 1977.
- 10 Saunders MacLane and Ieke Moerdijk. *Sheaves in geometry and logic: A first introduction to topos theory*. Springer Science & Business Media, 2012.
- 11 Anders Mörtberg and Max Zeuner. Towards a formalization of affine schemes in cubical agda.
- 12 Amnon Neeman. *Algebraic and analytic geometry*. Number 345. Cambridge University Press, 2007.
- 13 Jean-Pierre Serre. Géométrie analytique et géométrie algébrique. *Ann. Inst. Fourier, VI (1955–56)*, pages 1–42, 1955.
- 14 The Stacks Project Authors. *Stacks Project*. <https://stacks.math.columbia.edu>, 2018.
- 15 Ravi Vakil. The rising sea: Foundations of algebraic geometry. 2017. URL <http://virtualmath1.stanford.edu/~vakil/216blog>, 24:29.
- 16 Vladimir Voevodsky, Benedikt Ahrens, Daniel Grayson, et al. UniMath — a computer-checked library of univalent mathematics. available at <http://unimath.org>. URL: <https://github.com/UniMath/UniMath>.
- 17 Eric Wieser and Jujian Zhang. Graded rings in lean’s dependent type theory. In *International Conference on Intelligent Computer Mathematics*, pages 122–137. Springer, 2022.